

Семафори

- Семафор је ненегативна целобројна променљива на којој су дефинисане операције:
- wait(s) испитује да ли вредност семафора s задовољава услов s>0 и ако је услов задовољен онда се декрементира вредност семафора (s=s-1), а ако није, онда се чека док се тај услов не испуни. Провера задовољености услова s>0 и декрементирање вредности семафора се обављају атомски.
- **signal**(*s*) инкрементира вредност семафора (*s*=*s*+*1*). Инкрементирање вредности семафора се обавља атомски.
- init(s, val) поставља почетну вредност (s = val).

Семафори – једна имплементација

wait(s)

if s>0 then

s=s-1

else begin

Заустави процес и стави га у стање чекања код ОС

Стави процес у ред чекања на семафору ѕ

Ослободи процесор

end

Семафори – једна имплементација

signal(s)

if ред чекања на семафору s празан then


s=s+1


else begin

Уклони процес из реда чекања на семафору ѕ

Активирај тај процес код ОС – стави га у ред спремних

end





Два типа процеса, читаоци и писци, приступају једном запису (у општем случају запис може припадати некој колекцији података - бази података, датотеци, низу, уланчаној листи, табели итд.) Читаоци само читају садржај записа, а писци могу да читају и мењају садржај записа. Да не би дошло до нерегуларне ситуације у којој запису истовремено приступа више писаца или истовремено приступају и писци и читаоци, писци имају право ексклузивног приступа. Са друге стране, дозвољено је да више читалаца истовремено приступа запису (нема ограничења у њиховом броју). Написати програм којим се реализује рад процеса читалаца и писаца.

```
program Readers_Writers;
var db, mutexR : semaphore;
   readerCount : integer;
procedure Reader(ID : integer);
begin
   while (true) do
   begin
         wait(mutexR);
          readerCount := readerCount + 1;
         if (readerCount = 1) then wait(db);
         signal(mutexR);
         readData();
         wait(mutexR);
          readerCount := readerCount - 1;
         if (readerCount = 0) then signal(db);
         signal(mutexR);
   end
end:
```

```
procedure Writer(ID : integer);
begin
   while (true) do
   begin
          createData();
          wait(db);
          writeData();
          signal(db);
   end;
end;
begin
   init (db, 1);
   init(mutexR, 1);
   readerCount := 0;
   cobegin
          Writer(0);
          Reader(0);
   coend;
```

end.

```
program Readers_Writers;
var db, mutexR, in : semaphore;
   readerCount : integer;
procedure Reader(ID : integer);
begin
   while (true) do
   begin
         wait(in);
          wait(mutexR);
          readerCount := readerCount + 1;
          if (readerCount = 1) then wait(db);
          signal(mutexR);
          signal(in);
          readData();
          wait(mutexR);
          readerCount := readerCount - 1;
          if (readerCount = 0) then signal(db);
          signal(mutexR);
   end
end:
```

```
procedure Writer(ID : integer);
begin
   while (true) do
   begin
          createData();
          wait(in);
          wait(db);
          writeData();
          signal(db);
          signal(in);
   end;
end;
begin
   init (db, 1);
   init(mutexR, 1);
   readerCount := 0;
   init (in, 1);
   cobegin
          Writer(0);
          Reader(0);
   coend;
```


end



```
program Producer_Consumer;
const BufferSize = 3;
var mutex : semaphore;
   empty: semaphore;
   full: semaphore;
procedure Producer(ID : integer);
var item : integer;
begin
   while (true) do
   begin
         make_new(item);
         wait(empty);
         wait(mutex);
         put_item(item);
         signal(mutex);
         signal(full);
   end:
end:
```

```
procedure Consumer(ID : integer);
var item : integer;
begin
    while (true) do
    begin
        wait(full);
        wait(mutex);
        remove_item(item);
        signal(mutex);
        signal(empty);
        consume_item(item);
    end;
end;
```

```
begin
    init(mutex,1);
    init(empty, BufferSize);
    init(full, 0);
    cobegin
        Producer(0);
        ...
        Consumer(0);
        ...
    coend;
end.
```



Постоји један произвођач и N потрошача који деле заједнички једноелементни бафер. Произвођач убацује производ у бафер и чека док свих N потрошача не узму исти тај производ. Тада започиње нови циклус производње.

```
program AtomicBroadcast;
const
         N = 5;
var mutex : semaphore;
   empty: semaphore;
   full: array [1..N] of semaphore;
   num : integer;
   index : integer;
procedure Producer;
var item, index : integer;
begin
   while (true) do
   begin
         wait(empty);
         make_new(item);
         for index := 1 to N do signal(full[index]);
   end;
end;
```

```
procedure Consumer(ID : integer);
var item : integer;
begin
   while (true) do
   begin
         wait(full[ID]);
         wait(mutex);
         get_item(item);
          num := num + 1;
          if (num = N) then
          begin
                   signal(empty);
                   num := 0;
         end;
         signal(mutex);
         consume_item(item);
   end;
end;
```

```
begin
    init(mutex,1);
    init(empty, 1);
    for index := 1 to N do init(full[index], 0);
    num := 0;
    cobegin
        Producer;
        Consumer(1);
        ...
        Consumer(N);
    coend;
end.
```

Постоји један произвођач и N потрошача који деле заједнички бафер капацитета В. Произвођач убацује производ у бафер на који чекају свих N потрошача, и то само у слободне слотове. Сваки потрошач мора да прими производ у тачно оном редоследу у коме су произведени, мада различити потрошачи могу у исто време да узимају различите производе.

```
program AtomicBroadcastB;
const N = 5;
B = 2;
var mutex : array [1..B] of semaphore;
empty : semaphore;
full : array [1..N] of semaphore;


buffer : array [1..B] of integer;
num : array [1..B] of integer;
readFromIndex : array [1..N] of integer;
writeToIndex : integer;
index : integer;
```

```
procedure put_item(var item : integer);
begin
   buffer[writeToIndex] := item;
   writeToIndex := (writeToIndex mod B) + 1;
end;
procedure Producer;
var item, index : integer;
begin
   while (true) do
   begin
          wait(empty);
          make_new(item);
          put_item(data);
          for index := 1 to N do signal(full[index]);
   end;
end;
```

```
procedure Consumer(ID : integer);
var item : integer;
begin
   while (true) do
   begin
         wait(full[ID]);
          wait(mutex[readFromIndex[ID]]);
          item := buffer[readFromIndex[ID]];
         //get_item(item, ID);
          num[readFromIndex[ID]] := num[readFromIndex[ID]] + 1;
          if (num[readFromIndex[ID]] = N) then
          begin
                   signal(empty);
                   num[readFromIndex[ID]] := 0;
          end;
          signal(mutex[readFromIndex[ID]]);
          readFromIndex[ID]:= (readFromIndex[ID] mod B) + 1;
         consume_item(item);
   end:
end;
```

```
begin
    init(mutex,1);
    init(empty, B);
    for index := 1 to N do init(full[index], 0);
    for index := 1 to N do readFromIndex[index] := 1;
    writeToIndex := 1;

cobegin
        Producer;
        Consumer(1);
        ...
        Consumer(N);
    coend;
end.
```



Пет филозофа седи око стола. Сваки филозоф наизменично једе и размишља. Испред сваког филозофа је тањир шпагета. Када филозоф пожели да једе, он узима две виљушке које се налазе уз његов тањир. На столу, међутим, има само пет виљушки. Значи, филозоф може да једе само када ниједан од његових суседа не једе. Написати алгоритам за филозофа (0 ≤ i ≤ 4).

```
program Dining_philosophers;
const n = 5;
var ticket: semaphore;
   fork: array [0..n-1] of semaphore;
   l: integer;
procedure think; begin ... end;
procedure eat; begin ... end;
```

```
procedure Philosopher(i : integer);
var left, right: integer;
begin
    left := i;
    right := (i + 1) \mod n;
    while (true) do
           begin
                      think;
                      wait (ticket);
                      wait (fork[left]);
                      wait (fork[right]);
                      eat;
                      signal (fork[right]);
                      signal (fork[left]);
                      signal (ticket)
           end
end;
```


```
begin
    init (ticket, n-1 );
    for i:=0 to n-1 do init(fork[i],1);
    cobegin
        Philosopher(0);
        Philosopher(1);
        Philosopher(2);
        Philosopher(3);
        Philosopher(4);
        coend
end.
```

```
program Dining_philosophers;
const n = 5;
var ticket: semaphore;
  fork: array [0..n-1] of semaphore;
  i: integer;
procedure think; begin ... end;
procedure eat; begin ... end;
```

```
procedure Philosopher(i : integer);
var firstodd, secondeven: 0..n-1;
begin
   if (i mod 2 = 1) then begin
          firstodd := i;
          secondeven := (i+1) mod n
   end
   else
   begin
          firstodd := (i+1) mod n;
          secondeven := i
   end;
   while (true) do
   begin
          think;
          wait(fork[firstodd]);
          wait(fork[secondeven]);
          eat:
          signal(fork[firstodd]);
          signal(fork[secondeven])
   end
end;
```

```
begin
    for i:=0 to n-1 do init(fork[i],1);
    cobegin
        Philosopher(0);
        Philosopher(1);
        Philosopher(2);
        Philosopher(3);
        Philosopher(4)
        coend
end.
```

The dining savages problem



The dining savages problem

Племе људождера једе заједничку вечеру из казана који може да прими М порција куваних мисионара. Када људождер пожели да руча, онда се он сам послужи из заједничког казана, уколико казан није празан. Уколико је казан празан, људождер буди кувара и сачека док кувар не напуни казан. Није дозвољено будити кувара уколико се налази бар мало хране у казану. Користећи семафоре написати програм који симулира понашање људождера и кувара.


The dining savages problem


```
program DiningSavages;
const M =...;
         cook: semaphore;
var
         savager: semaphore;
         mutex: semaphore;
         servings: shared integer;
procedure PrepareLunch; begin ...end
procedure GetServingFromPot; begin ...end
procedure SavageCook;
begin
   while (true) do
   begin
         wait (cook);
         PrepareLunch;
         servings := M;
         signal (savager)
   end;
end;
```

The dining savages problem

```
procedure Savage(i : integer);
begin
   while (true) do
   begin
          wait(mutex);
          if (servings = 0) then
          begin
                    signal (cook);
                    wait (savager);
          end;
          servings := servings - 1;
          GetServingFromPot;
          signal (mutex);
          eat:
   end;
end;
```

The dining savages problem





Постоји N пчела и један гладан медвед. Они користе заједничку кошницу. Кошница је иницијално празна, а може да прими Н напрстака меда. Медвед спава док се кошница не напуни медом, када се напуни медом, он поједе сав мед након чега се враћа на спавање. Пчелице непрестано лете од цвета до цвета и сакупљају мед. Када прикупе један напрстак долазе и стављају га у кошницу. Она пчела која је попунила кошницу буди медведа. Користећи семафоре решити проблем.

```
program BearAndHoneybees;
         N = ...;
const
         H = ...;
         hive : semaphore;
var
         full: semaphore;
          pot : shared integer;
procedure Bear;
procedure sleep; begin ... end;
procedure eat; begin ... end;
begin
   while (true) do
   begin
         sleep;
         wait(full);
         eat;
          pot := 0;
         signal(hive);
   end:
end;
```

```
procedure Honeybee (id : integer);
procedure collect; begin ... end;
begin
   while (true) do
   begin
          collect:
          wait(hive);
          pot := pot + 1;
          if (pot = H) then signal(full)
          else signal(hive);
   end:
end;
begin
   pot := 0;
   init(hive, 1);
   init(full, 0);
   cobegin
          Bear:
          Honeybee(1);
          Honeybee(2);
   coend;
```


end.

Разматра се проблем синхронизације на баријери (Barrier Synchronization). Синхронизациона баријера омогућава нитима да на њој сачекају док тачно N нити не достигне одређену тачку у извршавању, пре него што било која од тих нити не настави са својим извршавањем. Користећи семафоре решити овај проблем. Омогућити да се иста баријера може користити већи број пута.

```
program Barrier(input, output);
const
        N = ...;
var barrier1, barrier2 : Semaphore;
   cnt: Integer;
procedure Pass(id : integer);
begin
   wait(barrier1);
   cnt := cnt + 1;
   if(cnt = N) then
          signal(barrier2)
   else
          signal(barrier1);
   wait(barrier2);
   cnt := cnt - 1;
   if(cnt = 0) then
          signal(barrier1)
   else
          signal(barrier2);
end;
```

```
begin
    init(barrier1, 1);
    init(barrier2, 0);
    cnt := 0;

cobegin
        Pass(0);
        Pass(1);
        ...;
    coend
end.
```



Постоје два типа атома, водоник и кисеоник, који долазе до баријере. Да би се формирао молекул воде потребно је да се на баријери у истом тренутку нађу два атома водоника и један атом кисеоника. Уколико атом кисеоника дође до баријере на којој не чекају два атома водоника, онда он чека да се они сакупе. Уколико атом водоника дође до баријере на којој се не налазе један кисеоник и један водоник, он чека на њих. Баријеру треба да напусте два атома водоника и један атом кисеоника. Користећи семафоре написати програм који симулира понашање водоника и кисеоника.

```
program H2O;
var
   hydroSem : semaphore;
   hydroSem2 : semaphore;
   hydroMutex : semaphore;
   oxySem : semaphore;
   oxyMutex : semaphore;
   count: integer;
procedure Oxygen(i : integer);
begin
   wait (oxyMutex);
   signal (hydroSem);
   signal (hydroSem);
   wait (oxySem);
   bond (i);
   signal (oxyMutex);
end;
```

```
procedure Hydrogen(i : integer);
begin
   wait (hydroSem);
   wait (hydroMutex);
   count := count + 1;
   if (count = 2) then
   begin
         signal (oxySem);
         signal (hydroSem2);
         signal (hydroSem2);
         count := 0
   end;
   signal (hydroMutex);
   wait (hydroSem2);
   bond (i)
end;
```

```
begin
   init(hydroSem, 0);
   init(hydroSem2, 0);
   init(hydroMutex, 1);
   init(oxySem, 0);
   init(oxyMutex, 1);
   count := 0;
   cobegin
         Oxygen(1);
         Oxygen(2);
         Hydrogen(1);
         Hydrogen(2);
   coend;
end.
```

```
program H2O;
type Molecule = record ids : array[0..2] of integer end;


var
    hydroSem : semaphore;
    oxySem : semaphore;
    door1, door2 : semaphore;
    count : integer;
    mol : Molecule;

procedure bond(mol : Molecule); begin ... end;
```

```
procedure Oxygen(id : integer);
var mol : Molecule;
begin
   wait(oxySem);
   mol := barrier(id);
   bond(mol);
end;
procedure Hydrogen(id : integer);
var mol : Molecule;
begin
   wait(hydroSem);
   mol := barrier(id);
   bond(mol);
end;
```

```
function barrier(id : integer) : Molecule;
begin
   wait(door1);
   mol.ids[count] := id;
   count := count + 1;
   if (count = 3) then
    begin
          count := 0; signal(door2);
   end
   else
          signal(door1);
   wait(door2);
   barrier := mol;
   count := count + 1;
   if (count = 3) then
    begin
          count := 0;
          signal(oxySem); signal(hydroSem); signal(hydroSem);
          signal(door1);
   end
   else
          signal(door2);
                                                                                  54/69
end;
```

```
begin
   count := 0;
   init(hydroSem, 2);
   init(oxySem, 1);
   init(door1, 1);
   init(door2, 0);
   cobegin
          Oxygen(1);
          Oxygen(2);
          Hydrogen(1);
          Hydrogen(2);
   coend;
end.
```



У неком забавишту постоји правило које каже да се на свака три детета мора наћи барем једна васпитачица. Родитељ доводи једно или више деце у забавиште. Уколико има места оставља их, уколико не одводи их. Васпитачица сме да напусти забавиште само уколико то не нарушава правило. Написати процедуре, користећи семафоре, за родитеље који доводе и одводе децу и васпитачице и иницијализовати почетне услове.

```
program ChildCare;
const C = 3;
var numChild : integer;
   numNann: integer;
   numWaiting: integer;
   mutex : semaphore;
   confirm: semaphore;
   toLeave: semaphore;
function bringUpChildren (num: integer): boolean;
begin
   wait(mutex);
   if((numChild + num) <= C * numNann) then</pre>
   begin
         numChild := numChild + num;
         bringUpChildren := true;
   end
   else
         bringUpChildren := false;
   signal(mutex);
end:
```

```
procedure bringBackChildren (num : integer);
var out, i : integer;
begin
   wait(mutex);
   numChild := numChild - num;
   out : = numNann - (numChild + C - 1) / C;
   if(out > numWaiting) then out := numWaiting;
   for i := 0 to out do
   begin
          signal(toLeave);
          wait(confirm);
   end
   signal(mutex);
end;
```

```
procedure nannEnter();
begin
    wait(mutex);
    numNann := numNann + 1;
    if(numWaiting > 0) then
    begin
        signal(toLeave);
        wait(confirm);
    end
    signal(mutex);
end;
```

```
procedure nannExit();
begin
   wait(mutex);
   if (numChild <= C * (numNann - 1)) then</pre>
   begin
         numNann := numNann - 1;
         signal(mutex);
   end
   else
   begin
         numWaiting := numWaiting + 1
         signal(mutex);
         wait(toLeave);
         numNann := numNann - 1;
         numWaiting := numWaiting - 1
         signal(confirm);
   end
end;
```

```
begin
  numChild := 0;
  numNann := 0;
  numWaiting := 0;
  init(mutex, 1);
  init(confirm, 0);
  init(toLeave, 0);
  cobegin
  ...
  coend;
end.
```

Решити проблем уколико родитељ мора да сачека да остави децу све док не буде било места да их све остави.

```
program ChildCare;
const C = 3:
var numChild : integer;
   numNann: integer;
   numWaitingC: integer;
   numWaitingN: integer;
   mutex, entry, toLeave : semaphore;
procedure bringUpChildren (num : integer);
begin
   wait(entry)
   wait(mutex);
   if((numChild + num) > C * numNann) then
   begin
         numWaitingC := num;
         signal(mutex);
         wait(toLeave);
         numWaitingC :=0;
   end
   numChild := numChild + num;
   signal(mutex);
   signal(entry);
end;
```

```
procedure bringBackChildren (num : integer);
var out, i : integer;
begin
   wait(mutex);
   numChild := numChild - num;
   if ((numWaitingC > 0) and (numChild + numWaitingC <= C * numNann)) then
         signal(toLeave)
   else if ((numWaitingN > 0) and (numChild <= C * (numNann - 1))) then
         signal(toLeave);
   else
         signal(mutex);
end;
```

```
procedure nannEnter;
begin
    wait(mutex);
    numNann := numNann + 1;
    if ((numWaitingC > 0) and (numChild + numWaitingC <= C * numNann)) then
        signal(toLeave)
    else if ((numWaitingN > 0) and (numChild <= C * (numNann - 1))) then
        signal(toLeave);
    else
        signal(mutex);
end;</pre>
```

```
procedure nannExit;
begin
   wait(entry);
   wait(mutex);
   if (numChild > C * (numNann - 1)) then
   begin
         numWaitingN := numWaitingN + 1;
         signal(mutex);
         wait(toLeave);
          numWaitingN := numWaitingN - 1;
   end
   numNann := numNann - 1;
   signal(mutex);
   signal(entry);
end;
```

```
begin
  numChild := 0;
  numNann := 0;
  numWaitingC := 0;
  numWaitingN := 0;
  init(mutex, 1);
  init(entry, 1);
  init(toLeave, 0);
  cobegin
  ...
  coend;
end.
```

Питања?

Захарије Радивојевић, Сања Делчев Електротехнички Факултет Универзитет у Београду zaki@etf.rs, sanjad@etf.rs